
1 Gramatyki bezkontekstowe
Zadanie 1. Podaj gramatykę języka L = {w ∈ {a, b}∗ : #(a,w) = #(b, w)}.
Rozwiązanie zadania 1.

S ::= <eps> | aSbS | bSaS

Zadanie 2. Podaj gramatykę języka L = {w ∈ {a, b}∗ : #(a,w) 6= #(b, w)}.
Rozwiązanie zadania 2.

S ::= A | B
A ::= RaA | RaR
B ::= RbB | RbR
R ::= <eps> | aRbR | bRaR

Zadanie 3. Podaj gramatykę języka L = {w ∈ {a, b}∗ : #(a,w) = 2#(b, w)}.

Rozwiązanie zadania 3.

S ::= <eps> | SbSaSaS | SaSbSaS | SaSaSbS

Zadanie 4. Podaj gramatykę języka L poprawnych wyrażeń nawiasowych ().

Rozwiązanie zadania 4.

S ::= (S)S | <eps>

lub niejednoznaczana wersja:

S ::= (S) | SS | <eps>

Zadanie 5. Podaj gramatykę języka L poprawnych wyrażeń nawiasowych
()[], przy czym nawiasy kwadratowe nie mogą być zawarte w okrągłych.
Rozwiązanie zadania 5.

S ::= [S]S | (P)S | <eps>
P ::= (P)P | <eps>

Zadanie 6. Podaj gramatykę języka L = {w = anbkcm : n+ k = m}.
Rozwiązanie zadania 6.

S ::= aSc | X
X ::= bXc | <eps>

Zadanie 7. Podaj gramatykę języka L = {w = anbkcm : n+m = k}.
Rozwiązanie zadania 7.

S ::= XY
X ::= aXb | <eps>
Y ::= bYc | <eps>

Zadanie 8. Podaj gramatykę języka L = {w = anbkcm : n+m = k + 1}.
Rozwiązanie zadania 8.

S ::= aXY | XYc
X ::= aXb | <eps>
Y ::= bYc | <eps>

Zadanie 9. Podaj gramatykę języka L = {w = anbkcm : n+m 6= k}.
Rozwiązanie zadania 9.

S ::= XBY | AXY | XYC | AXYC
X ::= aXb | <eps>
Y ::= bYc | <eps>
A ::= aA | a
B ::= bB | b
C ::= cC | c

Zadanie 10. Podaj gramatykę języka L = {w = anbkcm : n+ 2m = k + 2}.
Rozwiązanie zadania 10. S ::= aaXY | XYc X ::= aXb | <eps> Y ::= bbYc
| <eps>

Zadanie 11. Wyrażenia arytmetyczne z prawostronnie łącznym potęgowa-
niem, czyli że np. x ∗ yzv

parsuje się jako x ∗ (y(zv))

Rozwiązanie zadania 11.

W ::= S | -S | W+S | W-S
S ::= C | S*C | S div C | S mod C
C ::= P | P^C
P ::= L | (W)
L ::= [1-9][0-9]* | 0

Zadanie 12. Podaj gramatykę języka L = {w ∈ {a, b}∗ : #(a,w) = 2k, k ≥
0}.
Rozwiązanie zadania 12.

S ::= aBaS | bS | <eps>
B ::= bB | <eps>

Zadanie 13. Podaj gramatykę języka L = {w ∈ {a, b}∗ : w 6= abababb}.
Rozwiązanie zadania 13.

S ::= aA | bX | <eps>
A ::= bB | aX | <eps>
B ::= aC | bX | <eps>
C ::= bD | aX | <eps>
D ::= aE | bX | <eps>
E ::= bF | aX | <eps>
F ::= bG | aX | <eps>
G ::= aX | bX
X ::= aX | bX | <eps>

Zadanie 14. Podaj gramatykę języka L = {w ∈ {a, b}∗ : w nie zawiera aa}.
Rozwiązanie zadania 14.

S ::= aB | B
B ::= bS | <eps>

Zadanie 15. Podaj gramatykę języka L = {w ∈ {a, b}∗ : w nie zawiera aaa}.

Rozwiązanie zadania 15.

S ::= aaB | aB | B
B ::= bS | <eps>

Zadanie 16. Podaj gramatykę języka L = {w ∈ {a, b}∗ : w nie zawiera abaab}.

Rozwiązanie zadania 16.

S ::= a S_1 | b S | <eps>
S_1 ::= b S_2 | a S_1 | <eps>
S_2 ::= a S_3 | b S | <eps>
S_3 ::= a S_4 | b S_2 | <eps>
S_4 ::= a S_1 | <eps>

Zadanie 17. Podaj gramatykę języka L = {w ∈ {a, b}∗ : w 6= anbnan}.
Rozwiązanie zadania 17.

S ::= A | AB | APQ | PBQ | PQB | ABABX | BX
P ::= aPb | ab
Q ::= bQa | ba
A ::= aA | a
B ::= bB | b
X ::= aX | bX | <eps>
Y ::= aX | bX

Zadanie 18. Przez segment słowa v ∈ A∗ składający się z tych samych liter x
nazwiemy takie słowo w = xk dla pewnego x ∈ A oraz k ∈ N+, że v = w′ww′′

dla pewnych w′, w′′ ∈ A∗ . Segment taki jest maksymalny, jeśli ani w′ nie kończy
się na x, ani w′′ nie zaczyna się od x. Napisz gramatykę generującą wszystkie
słowa v nad alfabetem {a, b, c} o tej właściwości, że każdy maksymalny segment
złożony z samych liter b w słowie v musi być bezpośrednio poprzedzony przez
dłuższy segment złożony z samych liter a.
Rozwiązanie zadania 18.

S ::= <eps> | C S | T S
T ::= A | a T b
A ::= a | aA
C ::= c | cC

Zadanie 19. Podaj gramatykę, która wygeneruje słowa nad alfabetem 0, 1,
które mają tyle samo jedynek na pozycjach parzystych co na pozycjach niepa-
rzystych.
Rozwiązanie zadania 19.

S ::= X | X 0 | X 01 X 1
X ::= <eps> | N X | A X B X | B X A X
N ::= 00 | 10
A ::= 01
B ::= 10

Zadanie 20. Napisz gramatykę generującą język takich słów nad alfabetem
{0, 1, 2}, w których pomiędzy każdymi dwoma zerami występuje jedynka, po-
między każdymi dwoma jedynkami występuje zero, (dwójki mogą występować
dowolnie), np. 1, 1010, 0221021.
Rozwiązanie zadania 20.

S ::= 2 S | 0 Y | 1 X | <eps>
X ::= 2 X | 0 Y | <eps>
Y ::= 2 Y | 1 X | <eps>

Zadanie 21. Podaj gramatykę, która wygeneruje słowa nad alfabetem a, b,
zawierające parzystą liczbę liter a i nieparzystą liczbę liter b.
Rozwiązanie zadania 21.

S ::= aZ | bX // parzyste a, nieparzyste b
X ::= aY | bS | <eps> // parzyste a, parzyste b
Y ::= aX | bZ // nieparzyste a, parzyste b
Z ::= aS | bY // nieparzyste a, nieparzyste b

